Code to program the Forecasting Tools in Wingridds V6

By José Manuel Gálvez

30 September 2025

To code the algorithms in Wingridds, you need three components: (1) code that needs to go inside the C:/WINGRIDDS/USER/ALIAS.USR file, where variables are programmed; (2) a code for the macro/script that generates the figure; and (3) a code for the corresponding color table.

Forecasting tool output should look like the following plots.

Galvez-Davison and Dynamics (GDID)



Trade Wind Inversion Characterization (TWIN)



Tropical Wave Tracking (TROPW)



Front and Airmass Characterization (FRAM)



Hail and Severe Weather Environments (GR02T)



The following files are idle for coding the algorithms in Wingridds version 6.

(1) ALIAS.USR Code for International Desk Algorithms

For the algorithms to work, you must have the variables needed available. To do so, you must copy the following code inside your C:/WINGRIDDS/USER/ALIAS.USR file. The code contains the definition of all the variables that will be used by the algorithms. Note that the following code is specific for four algorithms: TROPW, GDID, FRAM and TWIN. The code for GR02T to work should already be available inside your ALIAS.USR file.

```
@
@     TROPICAL WAVE ALGORITHM (TROPW) STARTS HERE
@
@
@   TROPW: WPC INTL DESKS TROPICAL WAVE TRACKING ALGORITHM
@   Developed by J.M.Galvez, S.Clarke, G.Chinchilla and
@   with collaboration of J.Alamo, B.Castellanos and
@   A.Levine (08-Jul-2024) - Last update 30-Sep-2025.
@
%    Correction factor for Vorticity for the South Hemisphere (TWLATMULT)
TWLATMULT=SADC 1 ZPOS SADC -2 ZNEG ALAT
%    Relative vor. (TWVOREL) and shear vor. (TWVOSHR) averaged in the
%    600-850 hPa layer, times 10^6. Which relates to Trop. Waves.
TWVOREL=SMLC 1000 SMLC 1000 SDVC 3 SSUM RVRT WIND 600 SSUM RVRT WIND 700 RVRT WIND 850
TWVOSHR=SMLC 1000 SMLC 1000 SDVC 3 SSUM WSHD 600 SSUM WSHD 700 WSHD 850
%    Curvature vor. in the same layer (TWVOCUR). In Wingridds, adding up TWVOREL and TWVOSHR
%    extracts the curvature vorticity, which is the only vorticity variable plotted in the algorithm.
TWVOCUR=SMLT TWLATMULT SSUM TWVOSHR TWVOREL
%    Moisture flux divergence averaged over 925-850-700 hPa, x 10^8.
%    COMA is the divergence of the flux of mixing ratio (DVRG FLUX MIXR WIND)
TWMC=SMLC 100 SMLC 1000 SMLC 1000 SDVC 3 SSUM COMA 925 SSUM COMA 850 COMA 700
%    Vector average of 850-700 winds in kt.
TWWN78=VAVR BKNT 700 BKNT 850
@
@     TROPICAL WAVE ALGORITHM (TROPW) ENDS HERE
@


@
@     GALVEZ-DAVISON INDEX AND DYNAMICS (GDID) BEGINS HERE
@
@
@   GDID: GDI AND DYNAMICS
@   Developed by J.M.Galvez - Last update 30-Sep-2025.
@
%    Low-Level winds, averaged in the 1000-850 hPa Layer (GDLWWND)
GDLWWND=VDVC 4 VSUM WIND 1000 VSUM WIND 950 VSUM WIND 900 WIND 850
%    Upper level winds, averaged in the 400-200 hPa Layer (GDUPWND)
GDUPWND=VDVC 4 VSUM WIND 400 VSUM WIND 300 VSUM WIND 250 WIND 200
%    Low-level moisture convergence in the 950-700 hPa layer (GDMC)
GDMC=SMTH SMTH SMLC 1+8 SDVC 4 SSUM COMA 950 SSUM COMA 900 SSUM COMA 850 COMA 700
%    Upper Divergence, average in the 400-200hPa layer (GDDDUU)
GDDV22=SAVR DVRG WIND 200 DVRG WIND 250
GDDV43=SAVR DVRG WIND 400 DVRG WIND 300
GDDDU1=SMTH SMTH SMTH SMTH SMLC 1+7 SAVR GDDV22 GDDV43
%    The followinc command truncates the field at a max of 250. If not truncated,
%    Wingridds does not allow to plot contours for certain values (Wingridds bug).
GDDDUU=SADC 250 ZPOS SSBC 250 GDDDU1
@
@     GALVEZ-DAVISON INDEX AND DYNAMICS (GDID) ENDS HERE
@


@
@     FRONT AND AIRMASS CHARACTERIZATION TOOL (FRAM) BEGINS HERE
@
@
@   FRAM: Front and Airmass Characterization Algorithm
@   Developed by J.M.Galvez - Last update 30-Sep-2025.
@
%    (1) Masks for lower versus higher terrain. FR_LMSK has 1 in terrain lower
%    than 1000m, while FR_HMSK does in the terrain higher than 1000m.
```

```
FR_LMSK=SNEG ZPOS SSBC 1 ZNEG SSBC 1000 HGHT SFC
FR_HMSK=SADC 1 ZPOS SSBC 1 ZNEG SSBC 1000 HGHT SFC
%===========================================================================
% (2) Calculation of Airmass field (FR_AIRM)
% (2.1) Thickness component of the airmass field (FR_THK)
FR_THK=SSUM SMLT FR_LMSK FR_LTHKF SMLT FR_HMSK FR_HTHKF
% (2.1.1) 1000-925 thickness for locations below 925 hPa. Divides by 23 GPM,
%  which is a value that approaches the standard deviation of the field.
FR_LTHK=SDVC 23 SDIF HGHT 925 HGHT 1000
FR_LTHKF=SMLT FR_LTHK FR_LTHK
% (2.1.2) 925-700 thickness, for locations above 925 hPa. Similarly, the
% field is divided by 80, value close to the field's standard deviation.
FR_HTHK=SDVC 80 SDIF HGHT 700 HGHT 925
FR_HTHKF=SMLT FR_HTHK FR_HTHK
% (2.2) Dewpoint component of the Airmass Field (FR_DW)
FR_DW=SMTH SSUM SMLT FR_LMSK FR_LDW SMLT FR_HMSK FR_HDW
% (2.2.1) Multiplies the dewpoint in 1000 and 925 hPa in Kelvin. Then
% divides by 100, which approaches the standard deviation of the field.
% for higher terrain, uses dewpoints of 850 and 925 hPa.
FR_LDW=SDVC 100 SMLT DWPK 1000 DWPK 925
FR_HDW=SDVC 100 SMLT DWPK 925 DWPK 850
% (2.3) Airmass Field (FR_AMF): Multiplies the dewpoint and thickness compo-
% nents. Divided by 50 and 58, which approach the st. deviation of the field.
FR_AMF=SMLT SDVC 50 FR_DW SDVC 58 FR_THK
%===========================================================================
% (3) Calculation of regions with airmass gradients.
% (3.1) Masks to handle terrain: FR_MSKG extracts gradients associated with
% terrain slopes. FR_MSKE extracts gradients in terrain above 3km. FR_MSTK
% maskouts regions where 1000-925 thickness exceed 673 GPM (too warm).
FR_MSKG=ZNEG SADC 1 SNEG ZNEG SSBC 10 SMLC 1000 MGRD HGHT SFC
FR_MSKE=SNEG ZPOS SSBC 1 ZNEG SSBC 3000 HGHT SFC
FR_MSTK=SADC 1 ZPOS SSBC 1 SNEG ZPOS SSBC 673 SMTH SDIF HGHT 925 HGHT 1000
FR_MSKS=SMLT FR_MSKG SMLT FR_MSKE FR_MSTK
% (3.2) Five factors used to calculate airmass gradients. The contribution
% weights are assigned by comparing against satellite and observations. The
% factors are 1000-925, 1000-850 and 1000-700 hPa thickness (FR_GR192,
% FR_GR185 and FR_GR170); 1000 hPa Thta-e (FR_THEFC) and the gradient of the
% airmass calculated in part 2 (FR_GRAIR=gradient magnitude of FR_AMF).
FR_GR192=SMLC 8+4 MGRD SDIF HGHT 925 HGHT 1000
FR_GR185=SMLC 5+4 MGRD SDIF HGHT 850 HGHT 1000
FR_GR170=SMLC 2+4 MGRD SDIF HGHT 700 HGHT 1000
FR_THEFC=SMLC 7+5 MGRD THTE 1000
FR_GRAIR=SMLC 7+4 MGRD FR_AMF
% (3.3) Airmass gradient (FR_GRADNT): Addition of the five gradient factors.
FR_GRADNT=SMLT FR_MSKS SSUM FR_THEFC SSUM FR_GRAIR SSUM FR_GR192 SSUM FR_GR185 FR_GR170
%===========================================================================
% (4) Moisture Convergence is calculated in the 1000-925 hPa layer, as it
% aids with surface boundary detection, and especially shear lines, which
% are not often associated with important thickness gradients.
% COMA=DVRG FLUX MIXR WIND, or divergence of the mix. ratio flux.
FR_MCOCOR=SMLT FR_MSKS SMTH SNEG SMLC 2+7 SAVR COMA 1000 COMA 925
@
@     FRONT AND AIRMASS CHARACTERIZATION TOOL (FRAM) ENDS HERE
@


@
@     TRADE WIND INTERSION CHARACTERIZATION (TWIN) BEGINS HERE
@
@
@ TWIN: TRADE WIND INVERSION HEIGHT/STRENGTH ALGORITHM
@ Developed by J.M.Galvez (5-Mar-2018)-Last update 30-Sep-2025.
@
% (1) Typical Lapse Rates in atmospheric layers in Celsius/50 hPa
TNTD1=2.5
TNTD2=3
% (2) Stability Assessment by evaluating the lapse rate in
% 50hPa layers starting from 1000 hPa through 650 hPa
TNIT10=ZNEG SADC TNTD1 SDIF TEMP 950 TEMP 1000
TNIC10=SMLC 975 SADC 1 ZPOS SSBC 1 TNIT10
TNIT95=ZNEG SADC TNTD1 SDIF TEMP 900 TEMP 950
TNIC95=SMLC 925 SADC 1 ZPOS SSBC 1 TNIT95
TNIT90=ZNEG SADC TNTD2 SDIF TEMP 850 TEMP 900
TNIC90=SMLC 875 SADC 1 ZPOS SSBC 1 TNIT90
TNIT85=ZNEG SADC TNTD2 SDIF TEMP 800 TEMP 850
TNIC85=SMLC 825 SADC 1 ZPOS SSBC 1 TNIT85
TNIT80=ZNEG SADC TNTD2 SDIF TEMP 750 TEMP 800
TNIC80=SMLC 775 SADC 1 ZPOS SSBC 1 TNIT80
TNIT75=ZNEG SADC TNTD2 SDIF TEMP 700 TEMP 750
TNIC75=SMLC 725 SADC 1 ZPOS SSBC 1 TNIT75
TNIT70=ZNEG SADC TNTD2 SDIF TEMP 650 TEMP 700
```

```
TNIC70=SMLC 675 SADC 1 ZPOS SSBC 1 TNIT70
%    (3) Inversion Level Calculation (TNILEV): Maximum pressure level where a
%    stable layer is present, in hPa; or pressure of the lowest stable layer.
%    (3.1) Definition of land (TNLMSK) and ocean (TNOMSK) masks.
TNLMSK=SADC 1 SNEG LAND SFC
TNOMSK=LAND SFC
%    (3.2) Calculation of the inversion over land.
TNILEVL=SMLT TNOMSK SMTH SMAX TNIC70 SMAX TNIC75 SMAX TNIC80 SMAX TNIC85 SMAX TNIC90 TNIC95
%    (3.3) Calculation of the inversion over the ocean.
TNILEVO=SMLT TNLMSK SMAX TNIC70 SMAX TNIC75 SMAX TNIC80 SMAX TNIC85 SMAX TNIC90 SMAX TNIC95 TNIC10
%    (3.4) Calculation of the inversion over the entire domain (final product).
TNILEV=SSUM TNILEVO TNILEVL
%    (4) Complimentary calculations:
%    (4.1) Relative humidity in the mid-troposphere (TNRELMID): Average 500-700 hPa
TNRELMID=SMTH SMTH SMTH SMTH SMTH SDVC 3 SSUM RELH 700 SSUM RELH 600 RELH 500
%    (4.2) Mid-tropospheric Winds (TNWINMID): Average 700-400 hPa
TNWINMID=VDVC 4 VSUM BKNT 700 VSUM BKNT 600 VSUM BKNT 500 BKNT 400
@
@        TRADE WIND INTERSION CHARACTERIZATION (TWIN) BEGINS HERE
@
```

(2) MACROS/SCRIPTS

The second step is to have the macros/algorithms inside the C:/WINGRIDDS/MACROS/ folder. Each macro has an independent code, and must be copied inside a file with extension ".CMD".

Galvez-Davison and Dynamics (GDID)

Filename: GDID.CMD

Code inside GDID.CMD:

```
LOOP
LNDF
# Legend -------------------------------------
LTX0 GDID: GALVEZ-DAVISON INDEX (GDI) AND DYNAMICS
LMN2 LFO3/LTM2 LLN2/
LTX3 GDI (shades) / 950-700hPa Moisture Convergence (green)
LTX4 1000-850hPa and 400-200hPa Averaged Flow [kt] (black and white)
LTX5 400-200hPa Divergence [10^-6 s^-1] (yellow)
FULG
# Background: Galvez-Davison Index (GDIF) ------
SMTH CTFC CFCGDID GDIF NOSCALE/
#---Low-level Moisture Convergence (Green contours)
THN2/
SADC 0.2 GDMC GRTN -5 LSTN -3 CIN1 CLR6 LNW1 NCLB DOTS/
SADC 0.4 GDMC GRTN -5 LSTN -3 CIN1 CLR6 LNW1 NCLB DOTS/
GDMC GRTN -5 LSTN -3 CIN1 CLRB LNW3 NCLB/
GDMC GRTN -5 LSTN -3 CIN1 CLR6 LNW1 NCLB/
GDMC GRTN -13 LSTN -11 CIN1 CLRF LNW5 NCLB/
GDMC GRTN -13 LSTN -11 CIN1 CLRB LNW3 NCLB/
GDMC GRTN -13 LSTN -11 CIN1 CLR6 LNW2 NCLB/
#---Low-level Moisture Divergence (Brown contours)
SSBC 0.2 SMTH GDMC GRTN +3 LSTN +5 CIN1 CLR8 LNW1 NCLB DOTS/
SSBC 0.4 SMTH GDMC GRTN +3 LSTN +5 CIN1 CLR8 LNW1 NCLB DOTS/
SMTH GDMC GRTN +3 LSTN +5 CIN1 CLR8 LNW3 NCLB/
SMTH GDMC GRTN +3 LSTN +5 CIN1 CLR0 LNW1 NCLB/
SMTH GDMC GRTN +11 LSTN +13 CIN1 CLRF LNW5 NCLB/
SMTH GDMC GRTN +11 LSTN +13 CIN1 CLR8 LNW3 NCLB/
SMTH GDMC GRTN +11 LSTN +13 CIN1 CLR0 LNW2 NCLB/
# ----Lat-Lon Grid-----------------------------------
NLBL NCLB SCL0 CIN5 GRTN -180 LSTN 182 SADC 0 LONG CLR3/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SADC 360 CLR3 LAST/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SSBC 720 CLR3 LAST/
NLBL NCLB SCL0 CIN5 LATI CLR3/
#---Low-level flow and barbs > 9kt----------------------
GDLWWND CLRD LNW1/STRM/
VMLC 2.2 GDLWWND GRTN 09 CLRD/BARB/
VMLC 2.2 GDLWWND GRTN 19 CLRF/
#---Upper Flow
GDUPWND CLR3 LNW1/GDUPWND CLRE LNW2/GDUPWND CLRD LNW3/STRM/
#---Upper Level Divergence----------------------------
GDDDUU CLRF GRTN 049 LSTN 051 CIN1 LNW5 NCLB/
GDDDUU CLR4 GRTN 049 LSTN 051 CIN1 LNW3 NCLB/
GDDDUU CLRF GRTN 119 LSTN 121 CIN2 LNW7 NCLB/
GDDDUU CLR4 GRTN 119 LSTN 121 CIN2 LNW3 NCLB/
GDDDUU CLR3 GRTN 119 LSTN 121 CIN2 LNW1 NCLB/
GDDDUU CLRF GRTN 239 LSTN 241 CIN2 LNW9 NCLB/
GDDDUU CLR4 GRTN 239 LSTN 241 CIN2 LNW4 NCLB/
GDDDUU CLR3 GRTN 239 LSTN 241 CIN2 LNW3 NCLB/
ENDL
```

Trade Wind Inversion Characterization (TWIN)

Filename: TWIN.CMD

Code inside TWIN.CMD:

```
LOOP
LNDF
LMN2 LFO3/LTM2 LLN2/
LTX0 TWIN: TRADE WIND INVERSION CHARACTERIZATION AND MID-LEVEL FLOW
LTX3 Inversion Base Height [hPa] (shades)
LTX4 Inversion strength [box:large=strong, small=weak, none=stable layer]
LTX5 700-400hPa Winds [kt] (beige) / 700-500hPa Rel. humidity [%] (pink/red)
LTX6 Galvez-Davison Index =30 and =40 (light blue)
#---Inversion height in shades----------------------------
CTFC CFCTWN SMTH TNILEV NOSCALE/
#---Small boxes to show weak inversion over the ocean-----
SMLT TNLMSK SMLC 10 SMTH TNIT10 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT95 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT90 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT85 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT80 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT75 GT15 BOX0 CLRD LNW1/
SMLT TNLMSK SMLC 10 SMTH TNIT70 GT15 BOX0 CLRD LNW1/
#---Small boxes to show weak inversion over the land-----
SMLT TNOMSK SMLC 10 SMTH TNIT95 GT15 BOX0 CLRD LNW1/
SMLT TNOMSK SMLC 10 SMTH TNIT90 GT15 BOX0 CLRD LNW1/
SMLT TNOMSK SMLC 10 SMTH TNIT85 GT15 BOX0 CLRD LNW1/
SMLT TNOMSK SMLC 10 SMTH TNIT80 GT15 BOX0 CLRD LNW1/
SMLT TNOMSK SMLC 10 SMTH TNIT75 GT15 BOX0 CLRD LNW1/
SMLT TNOMSK SMLC 10 SMTH TNIT70 GT15 BOX0 CLRD LNW1/
#---Large boxes to show strong inversion------------------
SMLT TNLMSK SMLC 10 SMTH TNIT10 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT95 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT90 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT85 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT80 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT75 GT35 BOX1 CLRF LNW2/
SMLT TNLMSK SMLC 10 SMTH TNIT70 GT35 BOX1 CLRF LNW2/
#---Small boxes to show weak inversion over the land-----
SMLT TNOMSK SMLC 10 SMTH TNIT95 GT35 BOX1 CLRF LNW2/
SMLT TNOMSK SMLC 10 SMTH TNIT90 GT35 BOX1 CLRF LNW2/
SMLT TNOMSK SMLC 10 SMTH TNIT85 GT35 BOX1 CLRF LNW2/
SMLT TNOMSK SMLC 10 SMTH TNIT80 GT35 BOX1 CLRF LNW2/
SMLT TNOMSK SMLC 10 SMTH TNIT75 GT35 BOX1 CLRF LNW2/
SMLT TNOMSK SMLC 10 SMTH TNIT70 GT35 BOX1 CLRF LNW2/
#---Relative Humidity in the Mid-Levels-------------------
TNRELMID GRTN 39 LSTN 41 CIN1 CLR9 DOTS LNW2/TNRELMID GRTN 39 LSTN 41 CIN1 CLR1 DOTS LNW3/
TNRELMID GRTN 24 LSTN 26 CIN1 CLR1 LNW2/TNRELMID GRTN 24 LSTN 26 CIN1 CLR2 LNW4/
TNRELMID GRTN 24 LSTN 26 CIN1 CLR9 LNW1/
TNRELMID GRTN 14 LSTN 16 CIN1 CLR1 LNW4/TNRELMID GRTN 14 LSTN 16 CIN1 CLRF LNW6/
TNRELMID GRTN 14 LSTN 16 CIN1 CLR2 LNW2/
#---Galvez-Davison Index----------------------------------
SMTH GDIF GRTN 29 LSTN 31 CIN2 CLR5 LNW1 NCLB/SMTH GDIF GRTN 29 LSTN 31 CIN2 CLR7 LNW3 NCLB/
SMTH GDIF GRTN 39 LSTN 41 CIN1 CLR5 LNW3 NCLB/SMTH GDIF GRTN 39 LSTN 41 CIN1 CLR7 LNW7 NCLB/
#---Lat-lon Grid------------------------------------------
NLBL NCLB SCL0 CIN5 GRTN -180 LSTN 182 SADC 0 LONG CLR3/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SADC 360 CLR3 LAST/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SSBC 720 CLR3 LAST/
NLBL NCLB SCL0 CIN5 LATI CLR3/
#---Mid-Level Wind----------------------------------
STRM TNWINMID CLRE LNW1/
TNWINMID GRTN 19 CLRE LNW1/
TNWINMID GRTN 34 CLR4 LNW1/
ENDL
```

Tropical Wave Tracking (TROPW)

Filename: TROPW.CMD

Code inside TROPW.CMD:

```
LNDF
LTX0 TROPW: WPC Intl Desks Tropical Wave Tracking Algorithm
LTM2 LLN2/
LTX3 Curvature Rel.Vort. 850-700-600 hPa (Shades)
LTX4 Pot.Vort averaged in the 320 and 325 isentropic surfaces (Yellow)
LTX5 850-700hPa Winds in kt (black) / Galvez-Davison Index > 25 (Red)
LTX6 Moisture flux convg. 925-850-700 hPa (Green Box)
# --- Curvature vorticity (shaded)  --------------
THN2/
CTFC CFCTWNHC SMTH SMTH SMTH SMTH SMTH SMTH TWVOCUR NOSCALE/
# --- Pot. Vorticity (yellow), calculated here ---------------------
SMTH SMLC 1+6 IPVO SLYR I320 I325 CLR4 LNW1 CIN5 GRTN 60 LSTN 95 NCLB DOTS/I320/
SMTH SMLC 1+6 IPVO SLYR I320 I325 CLR4 LNW2 CI10 GRTN 95 LSTN 105 NCLB/I320/
SMTH SMLC 1+6 IPVO SLYR I320 I325 CLR4 LNW1 CIN5 LSTN -60 GRTN -95 NCLB DOTS/I320/
SMTH SMLC 1+6 IPVO SLYR I320 I325 CLR4 LNW2 CI10 LSTN -95 GRTN -105 NCLB/I320/
# --- Lat-Lon Grid --------------------------------
NLBL NCLB SCL0 CIN5 GRTN -180 LSTN 182 LONG CLR3/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SADC  360 CLR3 LAST/
NLBL NCLB SCL0 CIN5 GRTN -179 LSTN 181 SSBC  720 CLR3 LAST/
NLBL NCLB SCL0 CIN5 LATI CLR3/
# --- Moisture Convergence (boxes) ----------------
SMTH TWMC LSTN -04 CLR6 BOX0 LNW2/
SMTH TWMC LSTN -10 CLR6 BOX1 LNW3/
SMTH TWMC LSTN -10 CLRB BOX0 LNW2/
# --- Galvez-Davison Index (red) ------------------
SMTH GDIF GRTN 19 LSTN 31 CIN5 NCLB LNW2 CLR2 DOTS/
SMTH GDIF GRTN 34 LSTN 49 CIN5 NCLB LNW3 CLR2/
SMTH GDIF GRTN 49 LSTN 56 CIN5 NCLB LNW6 CLR2/
# --- Winds (black) -------------------------------
STRM TWWN78 CLRD/
TWWN78 GRTN 10 CLRD LNW1/
TWWN78 GRTN 15 CLRF LNW1/
```

Front and Airmass Characterization (FRAM)

Filename: FRAM.CMD

Code inside FRAM.CMD:

```
LOOP
THN2/
LNDF
LTX0 FRAM: FRONT AND AIRMASS CHARACTERIZATION ALGORITHM
LMN2 LFO3/LTM2 LLN2/
LTX3 Airmass characteristics (shades)
LTX4 Airmass gradients (black, red and pink contours)
LTX5 1000-925 Thickness [GPM] (green/cyan)
LTX6 1000-925hPa Averaged Winds [kt] (dark gray)
# Background --------------------------------------
CTFC CFC_FRAM FR_AMF NOSCALE/
# Thickness 1000-925 -------------------------------
SMTH SDIF HGHT 925 HGHT 1000 CIN2 CLR6 LSTN 682 GRTN 650 NCLB DOTS/
SMTH SDIF HGHT 925 HGHT 1000 CIN1 CLR6 GRTN 650 LSTN 650 LNW2/
SMTH SDIF HGHT 925 HGHT 1000 CIN2 CLR5 LSTN 650 NCLB DOTS/
SMTH SDIF HGHT 925 HGHT 1000 CIN1 CLR5 GRTN 623 LSTN 623 LNW2/
# Moisture Convergence ------------------------------------
SMTH FR_MCOCOR BOX0 LNW2 GRTN 1.4 CLRD/
# Lat-Lon Grid -----------------------------------
NLBL NCLB SCL0 CI10 GRTN -180 LSTN 182 SADC 0 LONG CLR3/
NLBL NCLB SCL0 CI10 GRTN -179 LSTN 181 SADC  360 CLR3 LAST/
NLBL NCLB SCL0 CI10 GRTN -179 LSTN 181 SSBC  720 CLR3 LAST/
NLBL NCLB SCL0 CI10 LATI CLR3/
#Magnitude of gradients --------------------------
SMTH SMTH FR_GRADNT GRTN 24.5 LSTN 25.5 CIN1 NCLB CLRF LNW3/
SMTH SMTH FR_GRADNT GRTN 24.5 LSTN 25.5 CIN1 NCLB CLRD LNW1/
SMTH SMTH FR_GRADNT GRTN 39.5 LSTN 40.5 CIN1 NCLB CLRF LNW4/
SMTH SMTH FR_GRADNT GRTN 39.5 LSTN 40.5 CIN1 NCLB CLR2 LNW2/
SMTH SMTH FR_GRADNT GRTN 69.5 LSTN 70.5 CIN1 NCLB CLRF LNW5/
SMTH SMTH FR_GRADNT GRTN 69.5 LSTN 70.5 CIN1 NCLB CLR4 LNW2/
# Winds, plotted at a coarser resolution --------------
THN4/
VAVR BKNT 925 BKNT 1000 GT10 CLRD LNW1/
VAVR BKNT 925 BKNT 1000 GT25 CLRF LNW1/
STRM LAST CLRD/
# Terrain -------------------------------
FULG/
HGHT SFC GRTN 2500 CIN5 CLRD NCLB LNW7/
HGHT SFC GRTN 2500 CLRD BOX2 LNW7/
ENDL
```

Hail and Severe Weather Environments (GR02T)

Filename: GR02T.CMD

Code inside GR02T.CMD:

```
LOOP
LNDF
LTX0 GR02T (Galvez & Santayana, Dec 2019)
LTM2 LLN2/
LTX3 SEVERITY POTENTIAL SHADES: Gray=Strong Convection; Green=Marginal/Low
LTX4 Red=Low/Moderate; Fuscia=Elevated. Increases if boxes appear.
LTX5 POTENTIAL FOR HAIL: Red/fuscia boxes inside red/fuscia contours
LTX6 over a shaded area, mainly if Temp_500<-9C (light blue contours).
FULG/
CTFC CFCGRATS SMTH GRF1 NOSCALE/
# Grid ---------------------------------------------------------
THN2/
NLBL NCLB SCL0 CLRE CIN5 GRTN -180 LSTN 182 SADC 1 LONG DOTS/
NLBL NCLB SCL0 CLRE CIN5 GRTN -179 LSTN 181 SADC  360  LAST DOTS/
NLBL NCLB SCL0 CLRE CIN5 GRTN -179 LSTN 181 SSBC 720 LAST DOTS/
NLBL NCLB SCL0 CLRE CIN5 LATI DOTS/
# Low-Level Winds -------------------------------------------
LAST GRTN 38 CLR5/LAST GRTN 28 CLR6/VAVR BKNT 925 BKNT 850 GRTN 18 CLRB LNW1/
#-------Moisture Convergence green contours
SMLT GMF1 SMTH MCNV CLR6 GRTN -2.5 LSTN -1.5 CIN1 NCLB LNW1/
SMLT GMF1 SMTH MCNV CLR6 LSTN -2 GRTN -4 CIN1 DOTS NCLB/
#-------500 Temperatures
SMTH SMTH SMTH TEMP 500 CLR0 GRTN -8.5 LSTN -4.5 CIN1 LNW1 DOTS/
SMTH SMTH SMTH TEMP 500 CLR5 GRTN -9.5 LSTN -8.5 CIN1 LNW3/
SMTH SMTH SMTH TEMP 500 CLRC GRTN -16.5 LSTN -09.5 CIN1 LNW1/
SMTH SMTH SMTH TEMP 500 CLRC GRTN -20.5 LSTN -19.5 CIN1 LNW1/
#-------Upper Flow
THN4/
LAST GRTN 89 CLR3/VAVR BKNT 250 BKNT 200 GRTN 59 CLRE/
#----500 hPa Mixing Ratio contours
FULG/
SMLT GMF1 SMTH SMLC 1+4 MIXR 500 CLR7 NCLB GRTN 07.5 LSTN 08.5 CIN1 LNW2/
SMLT GMF1 SMTH SMLC 1+4 MIXR 500 CLR7 NCLB GRTN 19.5 LSTN 20.5 CIN1 LNW4/
#-------Green Hail Boxes
SMLT GMF1 GM_LR75 GT16 CLRF BOX1 LNW2/
SMLT GMF1 GM_LR75 GT16 CLR6 BOX0 LNW2/
SMLT GMF1 GM_LR75 GT20 CLRF BOX4 LNW2/
SMLT GMF1 GM_LR75 GT20 CLR6 BOX2 LNW2/
SMLT GMF1 GM_LR75 GT20 CLR6 BOX0 LNW3/
#-------Red Hail Boxes
SMLT GMF4 GM_LR75 GT16 CLRF BOX1 LNW2/
SMLT GMF4 GM_LR75 GT16 CLR2 BOX0 LNW2/
SMLT GMF4 GM_LR75 GT20 CLRF BOX4 LNW2/
SMLT GMF4 GM_LR75 GT20 CLR2 BOX2 LNW2/
SMLT GMF4 GM_LR75 GT20 CLR2 BOX0 LNW3/
#-------Magenta Hail Boxes
SMLT GMF5 GM_LR75 GT16 CLRF BOX1 LNW2/
SMLT GMF5 GM_LR75 GT16 CLR1 BOX0 LNW2/
SMLT GMF5 GM_LR75 GT20 CLRF BOX4 LNW2/
SMLT GMF5 GM_LR75 GT20 CLR1 BOX2 LNW2/
SMLT GMF5 GM_LR75 GT20 CLR1 BOX0 LNW3/
#-------Red and Magenta Hail Contours
SMLT GMKTT SMLC 10 SMTH GMF2 GRTN 0.9 LSTN 1.1 CIN1 CLRF NCLB LNW5/
SMLT GMKTT SMLC 10 SMTH GMF2 GRTN 0.9 LSTN 1.1 CIN1 CLR2 NCLB LNW2/
SMLT GMKTT SMLC 10 SMTH GMF3 GRTN 0.9 LSTN 1.1 CIN1 CLRF NCLB LNW6/
SMLT GMKTT SMLC 10 SMTH GMF3 GRTN 0.9 LSTN 1.1 CIN1 CLR1 NCLB LNW2/
#-------Upper Divergence Contours
THN2/
DDU3 CLRF GRTN 149.5 LSTN 150.5 CIN1 LNW6 NCLB/DDU3 CLRF GRTN 79.5 LSTN 80.5 CIN1 LNW4 NCLB/
DDU3 CLR4 GRTN 149.5 LSTN 150.5 CIN1 LNW3 NCLB/DDU3 CLR4 GRTN 79.5 LSTN 80.5 CIN1 LNW2 NCLB/
DDU3 CLRF GRTN 289.5 LSTN 290.5 CIN1 LNW8 NCLB/
DDU3 CLR4 GRTN 289.5 LSTN 290.5 CIN1 LNW4 NCLB/
ENDL
```

(3) ASSOCIATED COLOR TABLES

The third component of the coding of the forecasting tools are the color tables associated with each tool. They need to be placed inside the directory C:/WINGRIDDS/USER/ . Each color table needs to have the extension .DAT.

GDID

Color Table Filename: CFCGDID.DAT

```
!
!Configuration file for Contour Line-Fill/Color
!Displays of Divergence
!
!Color Mode - 1=MAX/MIN COLOR, 2=MULTI INTERPOLATION, 3=INDIVIDUAL COLORS
!
3
!
!Number of Data Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
16
!
!Number of Color Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
17
!
!Min to Max Values
!
-10
-05
0
5
10
14
18
22
26
30
34
38
42
47
55
65
!
!R-G-B Values (Top-min, Bottom-max)
!
100 100 100
110 110 110
125 125 125
140 140 140
150 150 150
170 170 170
185 185 185
195 195 195
150 170 190
110 160 110
155 170 080
200 180 080
215 145 060
215 110 100
210 060 080
150 040 110
100 020 090
*****************************************************
*****************************************************
*************   END-OF-DATA   **********************
*****************************************************
*****************************************************
```

TWIN

Color Table Filename: CFCTWN.DAT

```
!
!Configuration file for Contour Line-Fill/Color
!
!Color Mode - 1=MAX/MIN COLOR, 2=MULTI INTERPOLATION, 3=INDIVIDUAL COLORS
!
3
!
!Number of Data Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
8
!
!Number of Color Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
8
!
!Min to Max Values
!
600
700
750
800
850
900
950
1000
!
!R-G-B Values (Top-min, Bottom-max)
!
070 060 080
110 050 160
073 080 135
055 115 065
080 135 060
160 130 060
155 097 060
120 067 060

*******************************************************
*******************************************************
************* END-OF-DATA ***********************
*******************************************************
*******************************************************
```

TROPW

Color Table Filename: CFCTWNHC.DAT

```
!
!Configuration file for Contour Line-Fill/Color
!Displays of Cloud Cover
!
!Color Mode - 1=MAX/MIN COLOR, 2=MULTI INTERPOLATION, 3=INDIVIDUAL COLORS
!
3
!
!Number of Data Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
5
!
!Number of Color Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
6
!
!Min to Max Values
!
1
10
20
30
40
!
!R-G-B Values (Top-min, Bottom-max)
!
130 130 130
170 150 120
190 170 000
225 150 030
205 130 095
205 110 125
*******************************************************
*******************************************************
*************   END-OF-DATA   ***********************
*******************************************************
*******************************************************
```

FRAM

Color Table Filename: CFC_FRAM.DAT

Content:

```
!
!Configuration file for Contour Line-Fill/Color
!Displays of Cloud Cover
!
!Color Mode - 1=MAX/MIN COLOR, 2=MULTI INTERPOLATION, 3=INDIVIDUAL COLORS
!
3
!
!Number of Data Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
41
!
!Number of Color Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
42
!
!Min to Max Values
!
174
176
178
180
182
184
186
188
190
192
194
196
198
200
202
204
206
208
210
212
214
216
218
220
222
224
226
228
230
232
234
236
238
240
242
244
246
248
250
252
254
256
258
!
!R-G-B Values (Top-min, Bottom-max)
!
185 185 200
175 175 195
165 165 190
155 155 185
145 145 180
140 140 180
160 150 160
155 145 155
150 140 150
144 134 144
136 126 136
128 118 128
120 110 120
140 110 130
130 105 135
120 100 139
110 095 143
100 090 147
```

```
094 092 152
093 099 156
092 106 160
091 113 164
090 120 168
070 130 154
071 135 153
072 145 152
073 155 151
074 165 150
077 162 120
080 157 094
105 164 092
130 168 090
150 172 088
188 170 082
186 161 085
184 152 088
182 143 091
180 130 104
178 120 106
176 115 110
174 110 114
172 098 095
*****************************************************
*****************************************************
*************   END-OF-DATA   ***********************
*****************************************************
*****************************************************
```

GR02T

Color Table Filename: CFCGRATS.DAT

```
!
!Configuration file for Contour Line-Fill/Color
!Displays of Cloud Cover
!
!Color Mode - 1=MAX/MIN COLOR, 2=MULTI INTERPOLATION, 3=INDIVIDUAL COLORS
!
3
!
!Number of Data Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
5
!
!Number of Color Entries (ONLY 2 ALLOWED IN COLOR MODE 1)
!
6
!
!Min to Max Values
!
4
30
60
120
220
!
!R-G-B Values (Top-min, Bottom-max)
!
062 061 055
098 098 104
115 115 123
090 140 068
190 093 082
230 115 210
*****************************************************
*****************************************************
*************   END-OF-DATA   ***********************
*****************************************************
*****************************************************
```